

Solution to Exercise 2: The QES-REC-CHK triplet

Key Point(s):

- For text fields it is frequently better to make them 'upper-case text' as 'm' is not the same as 'M', but even better is to avoid text fields altogether and to give preference to numeric coding with metadata in label blocks
- In the CHK file, it is always better to make all fields 'Must enter', except for automatically calculated variables. Then you will not have any missing values for the variable.

Tasks:

- o *Open the existing A_EX02.QES file and complete it.*
- o *Create the A_EX02.REC file (overwrite the existing one).*
- o *Edit the A_EX02.CHK file, make all fields MUSTENTER fields*
- o *Make Range and legal and label blocks where appropriate, but not both for the same variable.*
- o *Edit the field to ensure that label blocks are shown and add other key commands as appropriate such as AFTER ENTRY with IF ... THEN ... ENDIF statements.*

Solution:

The questionnaire would look as follows:

This is the questionnaire for the laboratory register

```
serno      Laboratory serial number #####   Enter 9001, 9002, ... if serial number
is not unique and write data entry note (F5)
regdate    Registration date <dd/mm/yyyy>   Enter 01/01/1800 if missing
sex        Examinee's sex #
age        Examinee's age in years ###      Enter 999 if missing
reason     Examination reason #
res1       Result of specimen 1 #
res1sc    Result of specimen 1 scanty #
res2       Result of specimen 2 #
res2sc    Result of specimen 2 scanty #
res3       Result of specimen 3 #
res3sc    Result of specimen 3 scanty #
```

EpiData Entry software is not case-sensitive, but the definition of values for text fields obviously is. Nevertheless, you make it best a habit to use lower-case for field names as some statistical packages are case-sensitive when it comes to field names.

or, if shown with “Preview data form”:

This is the questionnaire for the laboratory register

serno	Laboratory serial number	<input type="text"/>	Enter 9001, 9002, ... if serial number is not unique and write data entry note (F5)
regdate	Registration date	<input type="text"/>	Enter 01/01/1800 if missing
sex	Examinee's sex	<input type="text"/>	
age	Examinee's age in years	<input type="text"/>	Enter 999 if missing
reason	Examination reason	<input type="text"/>	
res1	Result of specimen 1	<input type="text"/>	
res1sc	Result of specimen 1 scanty	<input type="text"/>	
res2	Result of specimen 2	<input type="text"/>	
res2sc	Result of specimen 2 scanty	<input type="text"/>	
res3	Result of specimen 3	<input type="text"/>	
res3sc	Result of specimen 3 scanty	<input type="text"/>	

The *.REC file we created can be looked at in a text editor (like NotePad™ that comes with Windows™ or the powerful free text editor Crimson that is found on the course CD-ROM) and we see the following:

```
16 1 VLAB Filelabel: Exercise 2, Part A
_label1      1  1  30  0  0  0  0 112 This is the questionnaire for the laboratory register
#serno       1  3  30  37  3  0  4 112 serno      Laboratory serial number
_label2      41  3  30  0  0  0  0 112      Enter 9001, 9002, ... if serial number is not unique and write data entry not
_label3      121 3  30  0  0  0  0 112 e (F5)
_regdate     1  4  30  37  4  11 10 112 regdate     Registration date
_label4      49  4  30  0  0  0  0 112      Enter 01/01/1800 if missing
#sex         1  5  30  37  5  0  1 112 sex         Examinee's sex
#age         1  6  30  37  6  0  3 112 age         Examinee's age in years
_label5      40  6  30  0  0  0  0 112      Enter 999 if missing
#reason      1  7  30  37  7  0  1 112 reason      Examination reason
#res1        1  8  30  37  8  0  1 112 res1        Result of specimen 1
#res1sc      1  9  30  37  9  0  1 112 res1sc      Result of specimen 1 scanty
#res2        1  10 30  37  10 0  1 112 res2        Result of specimen 2
#res2sc      1  11 30  37  11 0  1 112 res2sc      Result of specimen 2 scanty
#res3        1  12 30  37  12 0  1 112 res3        Result of specimen 3
#res3sc      1  13 30  37  13 0  1 112 res3sc      Result of specimen 3 scanty
```

While this is perhaps not very informative to you at this point in time, you may note the simplicity of it. Have you ever tried to look at a spreadsheet in a text editor? You cannot, as it will not load and all you see is some gibberish. In contrast, this is a straight simple text file and its file size is just 1,382 bytes. In comparison, a Microsoft Word® 1997-2003 file containing the single letter “a”, nothing else, weighs in at 24,576 bytes...

For the *.CHK file, the entering of ranges and legal values took perhaps a bit trial and error. But basically it is very simple. For the field REGDATE we just entered:

```
01/01/2000-31/12/2005,01/01/1800
```

and for AGE

```
0-125,999
```

We can open the A_EX02.CHK file (CTRL+O, “Files of type”, “EpiData check file (*.chk)”) it is just a text file after all. It looks as follows:

```
LABELBLOCK
  LABEL label_sex
    1 Female
    2 Male
    9 "Sex not recorded"
  END
  LABEL label_reason
```

```

0 Diagnosis
1 "Follow-up at 1 month"
2 "Follow-up at 2 months"
3 "Follow-up at 3 months"
4 "Follow-up at 4 months"
5 "Follow-up at 5 months"
6 "Follow-up at 6 months"
7 "Follow-up at 7 months or later"
8 "Follow-up, month not stated"
9 "Reason not recorded"
END
LABEL label_result
0 Negative
1 "1+ positive"
2 "2+ positive"
3 "3+ positive"
4 "Positive, not quantified"
5 "Scanty, not quantified"
6 "Scanty, quantified"
9 "Result not recorded"
END
LABEL label_scanty
0 "Not applicable"
1 "1 AFB per 100 OIF"
2 "2 AFB per 100 OIF"
3 "3 AFB per 100 OIF"
4 "4 AFB per 100 OIF"
5 "5 AFB per 100 OIF"
6 "6 AFB per 100 OIF"
7 "7 AFB per 100 OIF"
8 "8 AFB per 100 OIF"
9 "9 AFB per 100 OIF"
END
END

serno
KEY UNIQUE 1
MUSTENTER
END

regdate
RANGE 01/01/2000 31/12/2005
LEGAL
01/01/1800
END
MUSTENTER
END

sex
COMMENT LEGAL USE label_sex SHOW
MUSTENTER
TYPE COMMENT
END

age
RANGE 0 125
LEGAL
999
END
MUSTENTER
END

```

```

reason
  COMMENT LEGAL USE label_reason SHOW
  MUSTENTER
  TYPE COMMENT
END

res1
  COMMENT LEGAL USE label_result SHOW
  MUSTENTER
  TYPE COMMENT
  AFTER ENTRY
    IF res1<>6 THEN
      res1sc=0
      GOTO res2
    ENDIF
  END
END

res1sc
  COMMENT LEGAL USE label_scanty SHOW
  MUSTENTER
  TYPE COMMENT
  AFTER ENTRY
    IF (res1=6) AND (res1sc=0) THEN
      HELP "Values of res1 and res1sc not compatible. Please verify"
      GOTO res1
    ENDIF
  END
END

res2
  COMMENT LEGAL USE label_result SHOW
  MUSTENTER
  TYPE COMMENT
  AFTER ENTRY
    IF res2<>6 THEN
      res2sc=0
      GOTO res3
    ENDIF
  END
END

res2sc
  COMMENT LEGAL USE label_scanty SHOW
  MUSTENTER
  TYPE COMMENT
  AFTER ENTRY
    IF (res2=6) AND (res2sc=0) THEN
      HELP "Values of res2 and res2sc not compatible. Please verify"
      GOTO res2
    ENDIF
  END
END

res3
  COMMENT LEGAL USE label_result SHOW
  MUSTENTER
  TYPE COMMENT
  AFTER ENTRY
    IF res3<>6 THEN

```

```

        res3sc=0
        GOTO WRITE
    ENDIF
END
END

res3sc
COMMENT LEGAL USE label_scanty SHOW
MUSTENTER
TYPE COMMENT
AFTER ENTRY
    IF (res3=6) AND (res3sc=0) THEN
        HELP "Values of res3 and res3sc not compatible. Please verify"
        GOTO res3
    ENDIF
END
END

```

Note that the capitalization versus the use of lower-case letters is used here only for easier visualization of the program flow (see note above about EpiData not being case-sensitive).

We will learn very soon how to edit the *.CHK file directly to experience its tremendous power.